Floyd's triangle   lists the natural numbers in a right triangle aligned to the left where

 the first row is   1     (unity)
 successive rows start towards the left with the next number followed by successive naturals listing one more number than the line above.


The first few lines of a Floyd triangle looks like this:

 1
 2  3
 4  5  6
 7  8  9 10
11 12 13 14 15



Task

 Write a program to generate and display here the first   n   lines of a Floyd triangle. 
(Use   n=5   and   n=14   rows).
 Ensure that when displayed in a mono-space font, the numbers line up in vertical columns as shown and that only one space separates numbers of the last row.


:: Floyd's triangle Task from Rosetta Code
:: Batch File Implementation
 
@echo off
rem main thing
setlocal enabledelayedexpansion
call :floydtriangle 5
echo(
call :floydtriangle 14
exit /b 0
 
:floydtriangle
set "fila=%1"
for /l %%c in (1,1,%fila%) do (
    set /a "lastRowNum=%%c+fila*(fila-1)/2"
    rem count number of digits of whole number trick
    rem source: https://stackoverflow.com/a/45472269
    set /a "Log=1!lastRowNum:~1!-!lastRowNum:~1!-0"
    set /a "numColum[%%c]=!Log:0=+1!"
)
echo(Output for %fila%
set "thisNum=1"
for /l %%r in (1,1,%fila%) do (
    set "printLine="
    for /l %%c in (1,1,%%r) do (
        rem count number of digits of whole number trick
        set /a "Log=1!thisNum:~1!-!thisNum:~1!-0"
        set /a "thisNumColum=!Log:0=+1!"
        rem handle spacing
        set "space= "
        set /a "extra=!numColum[%%c]!-!thisNumColum!"
        for /l %%s in (1,1,!extra!) do set "space=!space! "
        rem append current number to printLine
        set "printLine=!printLine!!space!!thisNum!"
        set /a "thisNum=!thisNum!+1"
    )
    echo(!printLine!
)
goto :EOF